

# Tudo em AdvPL

Compartilhando experiências e conhecimento em análise, programação e desenvolvimento.

## Algoritmos – Validação de CNH

Publicado em **01/05/2019**

### INTRODUÇÃO

Seguindo a sequência de algoritmos de validação de dígitos verificadores de documentos, vamos ver agora a numeração da CNH – Carteira Nacional de Habilitação.

### FORMATO

O número da CNH é formado de 11 dígitos, sendo os dois últimos dígitos verificadores, calculados a partir dos nove primeiros dígitos, usando algumas regras diferenciadas para o segundo dígito caso o módulo 11 do primeiro seja maior que 9. A aplicação do módulo 11 é muito parecida em todos os casos, porém as regras diferenciadas é que são o “pulo do gato”.

### FONTE EM ADVPL

```
#include 'protheus.ch'

/* -----
Referências
http://www.pctoledo.com.br/forum/viewtopic.php?f=43&t=6343
https://www.devmedia.com.br/forum/validacao-do-numero-de-registro-de-cnh
----- */

User Function TSTCNH()
Local cCNH
Local cDV

cCNH := "527988023"
cDV := DV_CNH(cCNH)

conout("CNH ..... "+cCNH)
conout("DV ..... "+cDV)

return
```

```

/* -----
Geração de dígito verificador de CNH
Recebe o número da CNG em String, com 9 dígitos
Calcula os dois dígitos verificadores
----- */

STATIC Function DV_CNH(cCNH)
Local nI , nVL
Local nM1 := 9 , nM2 := 1
Local nDV1 := 0 , nDV2 := 0
Local lMaior := .F.

// Calculo dos valores dos dígitos baseado
// no fator multiplicativo de acordo com a posição
// de cada dígito.

For nI := 1 to 9
    nVL := val( substr(cCNH,nI,1) )
    nDV1 += ( nVL * nM1 )
    nDV2 += ( nVL * nM2 )
    nM1--
    nM2++
    nVA := nVL
Next

// Determina o primeiro dígito verificador
nDV1 := nDV1 % 11
If nDV1 > 9
    nDV1 := 0
    lMaior := .T.
Endif

// Determina o segundo dígito verificador
nDV2 := nDV2 % 11
If lMaior
    // Regra diferenciada para caso o módulo 11
    // do primeiro dígito verificador seja maior que 9
    If nDV2 - 2 < 0
        nDV2 += 9
    ElseIf nDV2 - 2 >= 0
        nDV2 -= 2
    Endif
Endif

// Ajuste final
IF nDV2 > 9
    nDV2 := 0
Endif

```

```
// Retorna os dígitos calculados como string  
Return chr(48+nDV1) + chr(48+nDV2)
```

#### CONCLUSÃO

Por hora, acabei de concluir que a caixa de Pandora está aberta, e sobre esse assunto ainda vão sair vários posts !!!

**Agradeço a todos pela audiência e desejo a todos TERABYTES DE SUCESSO 😊**

#### REFERÊNCIAS

- <http://www.pctoledo.com.br/forum/viewtopic.php?f=43&t=6343>
- <https://www.devmedia.com.br/forum/validacao-do-numero-de-registro-de-cnh/354889>

Esse post foi publicado em **ADVPL, BLOG** e marcado **Programação** por **Júlio Wittwer**. Guardar link permanente [<https://siga0984.wordpress.com/2019/05/01/algoritmos-validacao-de-cnh/>]

5 RESPOSTAS EM "ALGORITMOS – VALIDAÇÃO DE CNH"



**Leonardo Nicolas**

em **05/12/2021 às 01:12** disse:

Valeu em! Confesso que no início fiquei perdido, mas depois eu fazendo na mão, me baseando em seu código, me encontrei de boas. 😊😊😊😊

Muito obrigado mesmo em! Esse seu conteúdo me ajudou a por em prática parte dos meus estudos. 😊😊😊😊

E até a mostrar ao meu padraço (caminhoneiro, mecânico e curioso. kkkkkk) como é que se faz validações de número de registro de CNH.....

★ Curtido por 1 pessoa



**Júlio Wittwer**

em **05/12/2021 às 15:12** disse:

Opa, eu que agradeço a audiência 😊 Que legal que o conteúdo foi útil 😊 !!!! 🤗

★ Curtir

**Rogério Lima**em **13/12/2021 às 12:56** disse:

Olá, fiz o teste com o num “22522791500” e disse ser inválido, está correto ?  
neste caso o dv calculado no 2º ato foi 8 e neste momento ele não atribuiria o (zero) .

★ Curtido por 1 pessoa**Júlio Wittwer**em **14/12/2021 às 23:37** disse:

Olá Rogério, tudo bem ?

O fonte de exemplo proposto calcula os 2 digitos verificadores finais da CNH, recebendo apenas os 9 primeiros digitos. Usando apenas os primeiros nove digitos, os dois ultimos deveriam ser “08”, e não “00”. Para usar a função de cálculo para fazer uma validação, você deve criar uma função que receba uma CNH com 11 digitos, separar os 9 primeiros, calcular o DV usando esse numero, e comparar o resultado calculado com os dois ultimos digitos informados 😊

Abraços e saudações !!!

★ Curtir

**Júlio Wittwer**em **14/12/2021 às 23:41** disse:

O algoritmo usado para a validação foi um porte para o AdvPL, partindo de dois exemplos que eu obtive em dois websites mencionados nas referências do post... fiz alguns testes com dois numeros de CNH ...risos ... não sei se existe alguma variação de cálculo que o código não “cobre” 😊

★ Curtir